
E85: Digital Design and Computer Engineering  
Problem Set 9   

Hint: review Section 7.3 of the textbook about the single-cycle processor until you 
are comfortable about how it operates and how to add new instructions. Appendix B 
defines the instructions.  

1) Suppose the RegWrite signal in a single-cycle RISC-V processor has a stuckat-
1 fault (i.e., the signal is always 1).  Which instructions would malfunction, 
and why?  

  
2) Modify the single-cycle RISC-V processor to implement the blt instruction.  

Mark up copies of the controller, main decoder, ALU decoder, and datapath 
(attached) to handle the new instruction as simply as possible.  Name any 
control signals you need to add.   

  
3) Modify the single-cycle RISC-V processor to implement the sll instruction.  

Mark up the Verilog (attached) to implement your changes as simply as 
possible.  

  
4) Alyssa P. Hacker is a crack circuit designer.  She offers to speed up one of the 

functional units in Table 7.7 by 50% (i.e., cut the delay in half) to improve the 
overall performance of the single-cycle processor.  Which unit should she 
optimize, and by what percentage will the execution time improve?  

  
5) Impact on Society: RISC-V has gained tremendous attention in recent years.  

Explain the market forces that have caused this interest.  
  

6) AI Question (Optional)  
This question must be solved by AI.  Report what the AI produces, whether you 
believe it is accurate or a hallucination, and whether the solution is similar, better, 
or worse than what you would have done yourself in a reasonable amount of time.    
  
Modify the single-cycle RISC-V processor from Digital Design and Computer 
Architecture to implement a load halfword instruction.  

  
How long did you spend on this problem set?  This will not count toward your 
grade but will help calibrate the workload.  
  

Problem 2: Controller  

Problem 2 ALU Decoder  
ALUOp  funct3  op5, funct75  ALUControl  Instruction  
00  XXX  XX  000 (add)  lw, sw  
01  XXX  XX  001 (subtract)  beq  
10  000  00, 01, 10  000 (add)  add  
10  000  11  001 (subtract)  sub  
10  010  XX  101 (set less than)  slt  
10  110  XX  011 (or)  or  
10  111  XX  010 (and)  and  

Problem 2 Main Decoder  
Instruction  Opcode  RegWrite  ImmSrc  ALUSrc  MemWrite  ResultSrc  Branch  ALUOp  Jump  
lw  0000011  1  00  1  0  01 0  00  0  
sw  0100011  0  01  1  1  XX  0  00  0  
R-type  0110011  1  XX  0  0  00  0  10  0  
beq  1100011  0  10  0  0  XX  1  01  0  
addi  0010011  1  00  0  0  00  0  10  0  
jal  0111111  1  11  X  0  10  0  XX  1  

Problem 3: Single-Cycle Processor Verilog  
  
module top(input  logic        clk, reset,  
           output logic [31:0] WriteData, DataAdr,  
           output logic        MemWrite); 
 
  logic [31:0] PC, Instr, ReadData; 
   
  // instantiate processor and memories 
  riscvsingle rvsingle(clk, reset, PC, Instr, MemWrite, DataAdr, WriteData, ReadData); 
  imem imem(PC, Instr); 
  dmem dmem(clk, MemWrite, DataAdr, WriteData, ReadData); 
endmodule 
 
module riscvsingle(input  logic        clk, reset, 
                   output logic [31:0] PC, 
                   input  logic [31:0] Instr, 
                   output logic        MemWrite, 
                   output logic [31:0] ALUResult, WriteData, 
                   input  logic [31:0] ReadData); 
 
  logic       ALUSrc, RegWrite, Jump, Zero; 
  logic [1:0] ResultSrc, ImmSrc; 
  logic [2:0] ALUControl; 
 
  controller c(Instr[6:0], Instr[14:12], Instr[30], Zero, ResultSrc, MemWrite, PCSrc, 
               ALUSrc, RegWrite, Jump, ImmSrc, ALUControl); 
  datapath dp(clk, reset, ResultSrc, PCSrc, ALUSrc, RegWrite, 
              ImmSrc, ALUControl, Zero, PC, Instr, ALUResult, WriteData, ReadData); 
endmodule 
 
module controller(input  logic [6:0] op, 
                  input  logic [2:0] funct3, 
                  input  logic       funct7b5, 
                  input  logic       Zero, 
                  output logic [1:0] ResultSrc, 
                  output logic       MemWrite, 
                  output logic       PCSrc, ALUSrc, 
                  output logic       RegWrite, Jump, 
                  output logic [1:0] ImmSrc, 
                  output logic [2:0] ALUControl); 
 
  logic [1:0] ALUOp; 
  logic       Branch; 
 
  maindec md(op, ResultSrc, MemWrite, Branch, ALUSrc, RegWrite, Jump, ImmSrc, ALUOp); 
  aludec  ad(op[5], funct3, funct7b5, ALUOp, ALUControl); 
 
  assign PCSrc = Branch & Zero | Jump; 
endmodule 
 
module maindec(input  logic [6:0] op, 
               output logic [1:0] ResultSrc, 
               output logic       MemWrite, 
               output logic       Branch, ALUSrc, 
               output logic       RegWrite, Jump, 
               output logic [1:0] ImmSrc, 
               output logic [1:0] ALUOp); 
 
  logic [10:0] controls; 
 
  assign {RegWrite, ImmSrc, ALUSrc, MemWrite, ResultSrc, Branch, ALUOp, Jump} = controls; 
 
  always_comb 
    case(op) 
    // RegWrite_ImmSrc_ALUSrc_MemWrite_ResultSrc_Branch_ALUOp_Jump 
      7'b0000011: controls = 11'b1_00_1_0_01_0_00_0; // lw 
      7'b0100011: controls = 11'b0_01_1_1_00_0_00_0; // sw 
      7'b0110011: controls = 11'b1_xx_0_0_00_0_10_0; // R-type  
      7'b1100011: controls = 11'b0_10_0_0_00_1_01_0; // beq 
      7'b0010011: controls = 11'b1_00_1_0_00_0_10_0; // I-type ALU 
      7'b1101111: controls = 11'b1_11_0_0_10_0_00_1; // jal 
      default:    controls = 11'bx_xx_x_x_xx_x_xx_x; // non-implemented instruction 
    endcase 
endmodule 
 
module aludec(input  logic       opb5, 
              input  logic [2:0] funct3, 
              input  logic       funct7b5,  
              input  logic [1:0] ALUOp, 
              output logic [2:0] ALUControl); 
 
  logic  RtypeSub; 
  assign RtypeSub = funct7b5 & opb5;  // TRUE for R-type subtract instruction 
 
  always_comb 
    case(ALUOp) 
      2'b00:                ALUControl = 3'b000; // addition 
      2'b01:                ALUControl = 3'b001; // subtraction 

      default: case(funct3) // R-type or I-type ALU 
                 3'b000:  if (RtypeSub)  
                            ALUControl = 3'b001; // sub 
                          else           
                            ALUControl = 3'b000; // add, addi 
                 3'b010:    ALUControl = 3'b101; // slt, slti 
                 3'b110:    ALUControl = 3'b011; // or, ori 
                 3'b111:    ALUControl = 3'b010; // and, andi 
                 default:   ALUControl = 3'bxxx; // ??? 
               endcase 
    endcase 
endmodule 
 
module datapath(input  logic        clk, reset, 
                input  logic [1:0]  ResultSrc,  
                input  logic        PCSrc, ALUSrc, 
                input  logic        RegWrite, 
                input  logic [1:0]  ImmSrc, 
                input  logic [2:0]  ALUControl, 
                output logic        Zero, 
                output logic [31:0] PC, 
                input  logic [31:0] Instr, 
                output logic [31:0] ALUResult, WriteData, 
                input  logic [31:0] ReadData); 
 
  logic [31:0] PCNext, PCPlus4, PCTarget; 
  logic [31:0] ImmExt; 
  logic [31:0] SrcA, SrcB; 
  logic [31:0] Result; 
 
  // next PC logic 
  flopr #(32) pcreg(clk, reset, PCNext, PC);  
  adder       pcadd4(PC, 32'd4, PCPlus4); 
  adder       pcaddbranch(PC, ImmExt, PCTarget); 
  mux2 #(32)  pcmux(PCPlus4, PCTarget, PCSrc, PCNext); 
  
  // register file logic 
  regfile     rf(clk, RegWrite, Instr[19:15], Instr[24:20], Instr[11:7], Result, SrcA, WriteData); 
  extend      ext(Instr[31:7], ImmSrc, ImmExt); 
 
  // ALU logic 
  mux2 #(32)  srcbmux(WriteData, ImmExt, ALUSrc, SrcB); 
  alu         alu(SrcA, SrcB, ALUControl, ALUResult, Zero); 
  mux3 #(32)  resultmux(ALUResult, ReadData, PCPlus4, ResultSrc, Result); 
endmodule 
 
module regfile(input  logic        clk,  
               input  logic        we3,  
               input  logic [ 4:0] a1, a2, a3,  
               input  logic [31:0] wd3,  
               output logic [31:0] rd1, rd2); 
 
  logic [31:0] rf[31:0]; 
 
  // three ported register file 
  // read two ports combinationally (A1/RD1, A2/RD2) 
  // write third port on rising edge of clock (A3/WD3/WE3) 
  // register 0 hardwired to 0 
 
  always_ff @(posedge clk) 
    if (we3) rf[a3] <= wd3;  
 
  assign rd1 = (a1 != 0) ? rf[a1] : 0; 
  assign rd2 = (a2 != 0) ? rf[a2] : 0; 
endmodule 
 
module adder(input  [31:0] a, b, 
             output [31:0] y); 
 
  assign y = a + b; 
endmodule 
 
module extend(input  logic [31:7] instr, 
              input  logic [1:0]  immsrc, 
              output logic [31:0] immext); 
  
  always_comb 
    case(immsrc)  
      2'b00:   immext = {{20{instr[31]}}, instr[31:20]};   // I-type       
      2'b01:   immext = {{20{instr[31]}}, instr[31:25], instr[11:7]};  // S-type (stores) 
      2'b10:   immext = {{20{instr[31]}}, instr[7], instr[30:25], instr[11:8], 1'b0}; // B-type (branches) 
      2'b11:   immext = {{12{instr[31]}}, instr[19:12], instr[20], instr[30:21], 1'b0}; // J-type (jal) 
      default: immext = 32'bx; // undefined 
    endcase              
endmodule 
 
module flopr #(parameter WIDTH = 8) 
              (input  logic             clk, reset, 
               input  logic [WIDTH-1:0] d,  
               output logic [WIDTH-1:0] q); 

  always_ff @(posedge clk, posedge reset) 
    if (reset) q <= 0; 
    else       q <= d; 
endmodule 
 
module mux2 #(parameter WIDTH = 8) 
             (input  logic [WIDTH-1:0] d0, d1,  
              input  logic             s,  
              output logic [WIDTH-1:0] y); 
 
  assign y = s ? d1 : d0;  
endmodule 
 
module mux3 #(parameter WIDTH = 8) 
             (input  logic [WIDTH-1:0] d0, d1, d2, 
              input  logic [1:0]       s,  
              output logic [WIDTH-1:0] y); 
 
  assign y = s[1] ? d2 : (s[0] ? d1 : d0);  
endmodule 
 
module imem(input  logic [31:0] a, 
            output logic [31:0] rd); 
 
  logic [31:0] RAM[63:0]; 
 
  initial 
      $readmemh("riscvtest.txt",RAM); 
 
  assign rd = RAM[a[31:2]]; // word aligned 
endmodule 
 
module dmem(input  logic        clk, we, 
            input  logic [31:0] a, wd, 
            output logic [31:0] rd); 
 
  logic [31:0] RAM[63:0]; 
 
  assign rd = RAM[a[31:2]]; // word aligned 
 
  always_ff @(posedge clk) 
    if (we) RAM[a[31:2]] <= wd; 
endmodule 
 
module alu(input  logic [31:0] a, b, 
           input  logic [2:0]  alucontrol, 
           output logic [31:0] result, 
           output logic        zero); 
 
  logic [31:0] condinvb, sum; 
 
  assign condinvb = alucontrol[0] ? ~b : b; 
  assign sum = a + condinvb + alucontrol[0]; 
 
  always_comb 
    case (alucontrol) 
      3'b000:  result = sum;       // add 
      3'b001:  result = sum;       // subtract 
      3'b010:  result = a & b;     // and 
      3'b011:  result = a | b;     // or 
      3'b101:  result = sum[31];   // slt 
      default: result = 32'bx; 
    endcase 
 
  assign zero = (result == 32'b0); 
endmodule  


